- **Source: The Transition to Chaos: Conservative Classical Systems and Quantum Manifestations, L. Reichl, 2nd edition, page 205**

Some basic definitions

```
In[20]:= randn[n_] := Table[RandomReal[NormalDistribution[0, 1]], {r, 1, n}, {c, 1, n}];
         symmetrize[mat_] := (mat + ConjugateTranspose[mat])/2;
         randGOE[n_] := symmetrize[randn[n]];
         randGUE[n_] := symmetrize[randn[n] + I randn[n]];
         realeigs[mat_] := Map[Re, Eigenvalues[mat]];
```

The functions ϕ, basically the Hermite polynomials multiplied by the Gaussian weight. For GOE eigenvalues, we will need the derivatives and integrals of these functions as well.

```
In[40]:= ϕ[n_, x_] := (Sqrt[π] 2^n n!)^(-1/2) Exp[-1/2 x^2] HermiteH[n, x];
         ϕderiv[n_, x_] := D[ϕ[n, y], y] /. y → x;
         ϕint[n_, x_] := Integrate[ϕ[n, t], {t, 0, x}];
```

The formula for the eignvalue density of the GUE is just $\sum_{k=0}^{N-1} \phi_k(x)^2$.

```
In[2]:= eigDensityGUE[N_, x_] := Sum[ϕ[k, x]^2, {k, 0, N - 1}];
```

The theoretical density for N=6...

```
In[31]:= Plot[eigDensityGUE[6, x], {x, -6, 6}]
```

Out[31]=



Compared with a random sample...

In[32]:=
```
Histogram[Flatten[Table[realeigs[randGUE[6]], {100 000}]]]
```

Out[32]=



Now the eigenvalue density for the GOE (for even N) is $\sum_{k=0}^{N/2-1}\left[\phi_{2k}(x)^2 - \phi'_{2k}(x)\int_0^x \phi_{2k}(t)\,\mathrm{dt}\right]$.

In[48]:=
```
eigDensityGOE[N_, x_] := Sum[ϕ[2 k, x]^2 - ϕderiv[2 k , x] ϕint[2 k , x], {k, 0, N / 2 - 1}];
```

The theoretical density for N=6...

In[59]:=
```
interpEigDensityGOE[n_] :=
  Interpolation[Table[{x, eigDensityGOE[n, x]}, {x, -6.0, 6.0, 0.05}]];
interpEigDensityGOE6 = interpEigDensityGOE[6];

Plot[interpEigDensityGOE6[x], {x, -6, 6}]
```

Out[57]=



Compared with a random sample...

In[58]:=
```
Histogram[Flatten[Table[realeigs[randGOE[6]], {100 000}]]]
```

Out[58]=